DEPARTMENT OF
MINERALS AND ENERGY

BUREAU OF MINERAL RESOURCES, GEOLOGY AND GEOPHYSICS

RECORD 1975/4

A FORTRAN PROGRAM FOR CALCULATING THE GRAVITY EFFECT OF A THREE-DIMENSIONAL BODY OF ARBITRARY SHAPE

by

B.R. Spies

RECORD 1975/4

A FORTRAN PROGRAM FOR CALCULATING THE GRAVITY EFFECT OF A THREE-DIMENSIONAL BODY OF ARBITRARY SHAPE

by

B.R. Spies

CONTENTS

ILLUSTRATIONS

2

SUMMARY

A description is given of a computer program used for calculating the gravity anomaly due to a three-dimensional body of arbitrary shape. The body is defined by a number of polygons representing depth contours of the body. By a system of interpolation between the contours the gravity anomaly caused by the three-dimensional body can be calculated to a high degree of precision.

All data are input on cards to a Cyber 76 computer and the output is a printed list of gravity values at stations along a profile or grid.

1. INTRODUCTION

Much work has been done on the computation of the gravity effects of two-dimensional bodies and computer programs for this purpose are available, for example Haigh, Pollard,& Williams (1972). It is not always possible to adapt such programs to simulate geological structures encountered in the field, owing to finite strike length or complex shape. Programs for simple three-dimensional bodies such as spheres and cylinders are available (Cull, 1971) but most existing methods for calculating the gravity effect of other three-dimensional bodies involve the use of graticules or mechanical integrators (Baranov, 1953).

The program described in this Record is based on a paper by Talwani & Ewing (1960) in which formulae are given for calculating the gravity effect of a three-dimensional body of arbitrary shape. The method basically involves dividing the body up into a large number of thin laminae, determining the gravity effect for each, and adding them. Horizontal laminae are used as they can be represented by contours of depth of the body.

2. MECHANISM OF COMPUTATION

The brief statement which follows in this section and in Appendix 1 is derived mainly from Talwani & Ewing (Op cit.).

The three-dimensional body is first represented by contours at various depths, each contour then being replaced by a horizontal irregular n-sided polygonal lamina. By making the number of sides n sufficiently large, the polygons can be made to approximate the contour lines as closely as desired. The gravity anomaly caused by each lamina is determined analytically by the subroutine GRACI and is stored as a function of the depth of the lamina (i.e. the contour elevation). By interpolation using numerical quadrature the subroutine NUQUAD fits a continuous curve relating the heights of the laminae with their gravity anomalies. The total area under this curve gives the gravity anomaly caused by the entire body.

The co-ordinate system is given in Appendix 1. The contour at depth z is replaced by a polygonal lamina of infitesimal thickness dz At point P the gravity anomaly due to the lamina is related to the solid angle it subtends and is given by the expression.

$$\Delta g = V\,dz \qquad \ldots\ldots (1)$$

where V is the anomaly caused by the lamina, per unit thickness. V can be expressed as two line integrals, both along the boundary of the polygon.

$$V = k\rho \left[\oint d\psi - \oint \frac{z}{(r^2+z^2)^{1/2}}\, d\psi \right] \qquad \ldots\ldots (2)$$

where k = universal constant of gravitation

ρ = volume density of the lamina

z, ψ and r = the cylindrical co-ordinates used to define the boundary of the polygon.

This expression can be simplified into components in the cartesian co-ordinate system shown in Appendix 1.

The total anomaly Δg_{total} caused by the entire body can be evaluated by integrating (1) between z top and z bottom, the vertical limits of the body.

$$\Delta g_{total} = \int_{z_{bottom}}^{z_{top}} V \, dz$$

To solve the integral the program stores V as a function of z, the depth of the contour, and by numerical quadrature fits a parabola through sets of three points. To an extent the closer the contours are spaced, the more accurate the determination of the anomaly.

The program assumes a constant density but could easily be modified to incorporate density variation with depth.

Accuracy

The accuracy of the method depends on how closely the irregular polygons fit the individual contours and on the precision of the numerical quadrature. The polygons can be made to fit the contours as closely as desired merely by increasing the number of sides but there is a corresponding increase in computer time. Obviously close fit to contours is only important when a portion of the contour boundary lies close to a point at which the computation is being made. The accuracy of the numerical quadrature is mainly governed by the contour interval. The assumption is made that V varies smoothly between contours and as long as the surface of the body varies smoothly between contours there is no advantage in interpolating depths between contours.

The most common source of error is caused by the area enclosed by a polygon differing significantly from that enclosed by the contour. If points are marked at intervals along a contour, and joined with straight lines, then the area enclosed by the polygon will in general be appreciably less than that enclosed by the contour. Care should therefore be taken to ensure that the area of the polygon is as close as possible to that of the contour.

3. DATA INPUT

All input data are on punched cards. Contours are drawn on the body and fitted with polygons, the corners of which are specified by X and Y co-ordinates. All distances are in metres.

First data card

This card contains a list of numbers specifying parameters of the body and traverse details.

NCONT is the number of contours defining the body, but can also include the top and/or bottom of the body. If the top and/or bottom of the body is not defined the program assumes the first and last contours define the extremities of the body, i.e., it has a flat top and/or bottom. In the general case when the top and/or bottom is not flat the depths are given on the third data card, and the top and/or bottom are defined as contours for the purpose of this data card.

NCONT must be an odd number less than 40.

DENS is the density contrast of the body in g/cm^3.

PX is the X co-ordinate of the beginning of the first profile.

PY is the corresponding Y co-ordinate.

DPX is the X increment for successive points on the profile.

DPY is the corresponding Y increment.

NPTS is the number of stations on the profile.

NLINES is the number of profiles.

DIST is the distance between profiles.

FIRST DATA CARD; NCONT, DENS, PX, PY, DPX, DPY, NPTS, NLINES, DIST

1 FORMAT: I2, F8.4, 4F10.2, 2I10, F10.2

Second data card

This gives a list of the number of co-ordinates in each contour, in order, from the top down. In the case of the top or bottom of the body the number may be 1. This represents a point and occurs for bodies with tops or bottoms which are not flat.

SECOND DATA CARD: NLIST (1), NLIST (2), NLIST (3),..................
2 FORMAT: I2, I2, I2,..............................

Third data card

This card gives a list of the depths to the contours from the top down. This includes the top and/or bottom of the body regardless of whether either of them are represented by a single point. 13 depths fit on a card. Extra cards may be used if there are more than 13 contours.

THIRD DATA CARD: DEPTH (1), DEPTH (2), DEPTH (3) ....
3 FORMAT: F6.1, F6.1, F6.1,.....................

Further data cards

There are two sets of data for each contour. The first set gives a list of X co-ordinates of the polygon, the second gives a list of corresponding Y co-ordinates in the same order. No data cards are required in the case of point contours. There may be up to 40 points for each contour, 13 fitting on a card.

```
DATA CARDS:  XLIST (1,1), XLIST (1,2), XLIST (1,3)...........
             YLIST (1,1), YLIST (1,2), YLIST (1,3)...........
             XLIST (2,1), XLIST (2,2), XLIST (2,3)...........
             YLIST (2,1), YLIST (2,2), YLIST (2,3)...........
                  .            .            .
                  .            .            .
                  .            .            .
FORMAT: F6.1, F6.1, F6.1,.......................
```

for both XLIST and YLIST

Further models

Data for new models may be inserted directly after the data for the previous model. A blank card should be the last card in the data deck to signify no further models.

Deck structure

See plate 4.

Data sheet format

See plate 3.

Program listing

See Appendix 2.

Program flow chart

See plate 1.

4. PROGRAM TESTING

The gravity profile over a sphere was computed in the following example. The sphere had the following parameters; depth to centre 500 m, radius 300 m, density contrast 0.5 g/cm^3. The exact gravity values are easily determined for this case, and the results (Plate 2) show that close fit is obtained by using only 3 contours with an 8-sided polygon defining them.
The top and bottom of the sphere were also defined.

The time taken on the Cyber 76 for the above model, which involved 11 observation points, was 1.9 seconds.

5. DISCUSSION

The program GRAV3D is capable of computing the gravity profile over three-dimensional bodies of complex shape. The accuracy depends on the number of contours employed and the number of sides defining the contour. However, increase in these parameters also increases computer time.

The program could be used for computing terrain corrections on a flat earth, and can be modified so that density variations with depth can be accommodated.

It was not considered warranted to include a PLOT facility in the program because of the extra computing time that would be involved.

6. REFERENCES

BARANOV, V. 1953 - Sur le calcul de l'influence gravimetrique des structures definies par les isobathes. Geophys. Prosp. 1(1), 36-43.

CULL, J.P., 1971 - A fortran program for calculating the gravity effect of vertical cylinders. Bur. Miner. Resour. Aust. Rec. 1971/74 (unpubl).

HAIGH, J.E., POLLARD, P.C., & WILLIAMS, J.P., 1972 - a computer program for calculation of gravity and magnetic curves for two-dimensional bodies of arbitrary cross-sections. Bur.Miner. Resour. Aust. Rec. 1972/49 (unpubl).

TALWANI, M., & EWING, M., 1960 - Rapid computation of gravitational attraction of three-dimensional bodies of arbitrary shape. Geophysics, 25(1), 203-25.

APPENDIX I. DERIVATION OF FORMULAE





P = point of computation

Z = depth to lamina of thickness dz representing contour

Z top = depth of top of body

Z bottom = depth of bottom of body

A.B.C.D.E.F = polygonal lamina

P' = projection of P on Plane A.B.C.D.E.F

r = radius vector in plane A.B.C.D.E.F

Gravity effect of lamina $V = k\rho \left[\oint d\psi - \oint \frac{z}{\sqrt{r^2 + z^2}} \, d\psi \right]$

where k = universal constant of gravitation

ρ = volume density of the lamina

z, ψ and r are cylindrical coordinates.

$$r = \frac{P_i}{\sin(\phi_i + \psi - \psi_{i+1})}$$

Second integral for segment BC = $\arcsin \frac{Z \cos \theta_i}{\sqrt{P_i^2 + Z^2}} - \arcsin \frac{Z \cos \phi_i}{\sqrt{P_i^2 + Z^2}}$

Total contribution from all sides is

$$V = k\rho \left[\sum_{i=1}^{n} \left\{ \psi_{i+1} - \psi_i - \arcsin \frac{Z \cos \theta_i}{\sqrt{P_i^2 + Z^2}} + \arcsin \frac{Z \cos \phi_i}{\sqrt{P_i^2 + Z^2}} \right\} \right]$$

APPENDIX 2

```
      PROGRAM GRAV3D (INPUT,OUTPUT,TAPE60=INPUT,TAPE61=OUTPUT)

C     THIS PROGRAM COMPUTES THE VERTICAL COMPONENT OF THE GRAVITATIONAL
C     EFFECT OF A 3-D BODY ALONG A SERIES OF PARALLEL PROFILES.
C
C     THE BODY IS DESCRIBED BY POLYGONS REPRESENTING DEPTH CONTOURS.
C
C     ALL DISTANCES ARE IN METRES.  DENSITY IS IN GMS/CC.
C
C     CONTOURS ARE LISTED FROM THE TOP DOWN.

      DIMENSION DEPTH(40),DELGA(40),XLIST(40,40),YLIST(40,40),NLIST(40)
    1 FORMAT(I2,F8.4,4F10.2,2I10,F10.2)
    2 FORMAT(40I2)
    3 FORMAT(13F6.1)
    4 FORMAT(*1MODEL NUMBER*I2/15H **************,3(/)* NCONT(NUMBER OF
     1CONTOURS) =*I3/* DENS(DENSITY OR DENSITY CONTRAST) =*F8.4*GMS/CC*/
     2* START OF FIRST PROFILE HAS COORDINATES PX =* F9.1*METRES,  PY =*
     3 F9.1*METRES*/* FOR SUCCESSIVE POINTS ALONG A PROFILE X INCREASES
     4BY DPX, =*F6.1*METRES, AND Y INCREASES BY DPY, =*F6.1*METRES*/* NP
     5TS(NUMBER OF POINTS ON EACH PROFILE) =*I2/* NLINES(NUMBER OF PROFI
     6LES TO BE COMPUTED) =*I2/* DIST(DISTANCE BETWEEN ADJACENT PROFILES
     7) =*F6.1*METRES*3(/))
    5 FORMAT( * CONTOUR NUMBER*I3* AT DEPTH*F7.1*METRES IS A POINT*/)
    6 FORMAT(* CONTOUR NUMBER*I3* AT DEPTH*F7.1 *METRES IS REPRESENTED B
     1Y*I3* POINTS AT COORDINATES* 3X*XLIST       YLIST*40(/,85X,F6.1,5X,
     2F6.1))
    8 FORMAT(*1*10X*PROFILE NUMBER*I3*, STARTING AT COORDINATES*2F8.1* W
     1ITH BEARING OF *F5.1* DEGREES FROM Y-AXIS*)
    9 FORMAT(*0*,15X,*X COORD*,11X,*Y COORD*,11X,*GRAVITY ANOMALY IN MGA
     1LS*)
   10 FORMAT(15X,F8.1,10X,F8.1,17X,F9.3)
      JOT=0
 1000 CONTINUE
      JOT=JOT+1

      READ 1,NCONT,DENS,PX,PY,DPX,DPY,NPTS,NLINES,DIST
C     FIRST DATA CARD GIVES THE FOLLOWING....
C           NCONT.....THE NUMBER OF CONTOURS REPRESENTING THE BODY. THIS
C                     INCLUDES ANY CONTOURS WHICH HAVE DEGENERATED TO A
C                     SINGLE POINT. NCONT MUST BE AN ODD NUMBER GREATER
C                     THAN 1 AND LESS THAN 0.
C           DENS......THE DENSITY OR DENSITY CONTRAST OF THE BODY.
C           PX,PY.....THE X AND Y STARTING COORDINATES OF THE FIRST
C                     PROFILE.
C           DPX,DPY...THE X AND Y INCREMENTS FOR SUCCESSIVE POINTS ON A
C                     PROFILE.
C           NPTS......THE NUMBER OF POINTS ON EACH PROFILE.
C           NLINES....THE NUMBER OF PROFILES REQUIRED.
C           DIST......THE DISTANCE BETWEEN ADJACENT PROFILES.

C     TEST TO SEE IF THERE ARE ANY FURTHER MODELS.
      IF(NCONT.EQ.0) GO TO 9999
      PRINT 4,JOT,NCONT,DENS,PX,PY,DPX,DPY,NPTS,NLINES,DIST

      READ 2,(NLIST(NL),NL=1,NCONT)
C     SECOND DATA CARD GIVES A LIST OF THE NUMBER OF SIDES OF EACH POLYGON


      READ 3,(DEPTH(ND),ND=1,NCONT)
C     THIS STATEMENT READS IN A LIST OF THE DEPTHS TO THE CONTOURS,
C           INCLUDING ANY WHICH HAVE DEGENERATED TO A SINGLE POINT

      DO 2003 I=1,NCONT
      NUM=NLIST(I)
      IF(NUM.EQ.1) PRINT 5,I,DEPTH(I)
      IF(NUM.EQ.1) GO TO 2003

      READ 3,(XLIST(I,IX),IX=1,NUM)
      READ 3,(YLIST(I,IY),IY=1,NUM)
C     THESE STATEMENTS READ IN SETS OF COORDINATES FOR THE APICES
C           OF EACH POLYGON.
C           NO COORDINATE PAIR MUST BE PRESENT FOR SINGLE POINT CONTOURS.

      IF(NUM.NE.1) PRINT 6,I,DEPTH(I),NLIST(I),(XLIST(I,IX),YLIST(I,IX),
     1IX=1,NUM)
 2003 CONTINUE
      NOTE=0
      NCONTL1=NCONT-1
C     THIS LOOP TESTS WHETHER OR NOT ANY OF THE CONTOURS ARE AT THE SAME
C           DEPTH. IF SO, SUBROUTINE ERROR1 PRINTS AN ERROR MESSAGE AND
C           THE MODEL IS ABANDONED.
      DO 99 I=1,NCONTL1
      IF(DEPTH(I).EQ.DEPTH(I+1))CALL ERROR1(I,DEPTH,NOTE)
   99 CONTINUE
      IF(NOTE.EQ.999) GO TO 1000
      DO 3904 I=1,NCONT
      NUM=NLIST(I)
      IF(NUM.EQ.1) GO TO 3904
C     THIS LOOP PUTS XLIST AND YLIST RELATIVE TO AN ORIGIN AT THE FIRST
C           POINT ON THE FIRST PROFILE.
      DO 2444 MM=1,NUM
      XLIST(I,MM)=XLIST(I,MM)-PX
      YLIST(I,MM)=YLIST(I,MM)-PY
 2444 CONTINUE
 3904 CONTINUE
      BEARNG=90.0
      IF(DPY.NE.0.0)BEARNG=ATAN(DPX/DPY)*180.0/3.1415927
      XU=SQRT(DIST**2/(DPX**2+DPY**2))
      DXXX=DPX*(NPTS-1)+DPY*XU
      DYYY=DPY*(NPTS-1)-DPX*XU
C     THIS LOOP MOVES TO THE NEXT LINE.
      DO 3901 N=1,NLINES
      IF(N.EQ.1) GO TO 2004
      PX=PX-XU*DPY
      PY=PY+XU*DPX
C     THESE LOOPS PUT XLIST AND YLIST RELATIVE TO THE NEW ORIGIN.
      DO 7650 I=1,NCONT
      NIT=NLIST(I)
      IF(NIT.EQ.1) GO TO 7650
      DO 7700 K=1,NIT
      XLIST(I,K)=XLIST(I,K)+DXXX
      YLIST(I,K)=YLIST(I,K)+DYYY
 7700 CONTINUE
 7650 CONTINUE
 2004 PRINT 8,N,PX,PY,BEARNG
      PRINT 9
      DO 3900 J=1,NPTS
      IF(J.EQ.1) GO TO 1234
C     THESE TWO LOOPS CAUSE AN INCREMENT OF ONE STATION ALONG THE PROFILE.
      DO 1001 II=1,NCONT
      NLIS=NLIST(II)
      IF(NLIS.EQ.1)GO TO 1001
      DO 1002 IX=1,NLIS
      XLIST(II,IX)=XLIST(II,IX)-DPX
      YLIST(II,IX)=YLIST(II,IX)-DPY
 1002 CONTINUE
 1001 CONTINUE
C     THIS LOOP USES SUBROUTINE GRA01 TO CALCULATE DELGA(N), THE
C           GRAVITATIONAL EFFECT OF EACH CONTOUR.
C           DELGA IS SET TO 0 IF THE CONTOUR IS A SINGLE POINT.
 1234 DO 5000 NC=1,NCONT
      NLISTN=NLIST(NC)
      IF(NLISTN.EQ.1) GO TO 4444
      Z=DEPTH(NC)
      CALL GRA01(NLIST,Z,XLIST,YLIST,TOTEQ,NC,DPX)
      DELGA(NC)=TOTEQ
      GO TO 5000
 4444 DELGA(NC)=0.0
 5000 CONTINUE
C     THE REMAINDER OF THE PROGRAM USES SUBROUTINE NUQUAD TO SUM THE
C           GRAVITATIONAL EFFECT OF THE CONTOURS AND PRINTS THE RESULT.
      CALL NUQUAD(NCONT,DEPTH,DELGA,TOTALG)
      GRAV=0.00667*DENS*TOTALG
      XX=PX+(J-1)*DPX
      YY=PY+(J-1)*DPY
      PRINT 10,XX,YY,GRAV
 3900 CONTINUE
 3901 CONTINUE
      GO TO 1000
 9999 STOP
      END
```

PROGRAM LISTING

APPENDIX 2

```
      SUBROUTINE NUQUAD(NCONT,DEPTH,DELGA,TOTALG)
C   THIS SUBROUTINE DOES A NUMERICAL QUADRATURE FOR SUMMING VALUES OF
C         DELGA AT DIFFERENT DEPTHS.
C    NCONT IS THE NUMBER OF CONTOURS
C   NCONT,DEPTH + DELGA ARE INPUT DATA.. TOTALG IS OUTPUT.
      DIMENSION DEPTH(40),DELGA(40)
      TOTALG=0.0
      IA=NCONT-2
      DO 6000 I=1,IA,2
      Z1=DEPTH(I)
      Z2=DEPTH(I+1)
      Z3=DEPTH(I+2)
      G1=DELGA(I)
      G2=DELGA(I+1)
      G3=DELGA(I+2)
      IF(G1.EQ.0..AND.G2.EQ.0.) G3=0.
      IF(G2.EQ.0..AND.G3.EQ.0.) G1=0.
      EQ9A=G1*(Z1-Z3)/(Z1-Z2)*(3*Z2-Z3-2*Z1)
      EQ9B=G2*(Z1-Z3)**3/(Z2-Z3)/(Z2-Z1)
      EQ9C=G3*(Z1-Z3)/(Z3-Z2)*(3*Z2-Z1-2*Z3)
      EQ9=(EQ9A+EQ9B+EQ9C)/6.0
      TOTALG=TOTALG+EQ9
 6000 CONTINUE
      RETURN
      END
```

```
      SUBROUTINE ERROR1(I,DEPTH,NOTE)
C    THIS SUBROUTINE PRINTS AN ERROR MESSAGE IF ANY ADJACENT CONTOURS ARE
C          AT THE SAME DEPTH, AND SETS NOTE=999 TO STOP EXECUTION OF
C           THE MODEL
      DIMENSION DEPTH(40)
    1 FORMAT(1X,4(/)*    ILLEGAL DATA ENTRY. CONTOURS*I3* AND*I3* ARE BOT
     1H AT DEPTH*F8.1)
      J=I+1
      PRINT 1,I,J,DEPTH(I)
      NOTE=999
      RETURN
      END
```

```
      SUBROUTINE GRAD1(NLIST,Z,XLIST,YLIST,TOTEQ,NC,DPX)
C    THIS SUBROUTINE COMPUTES THE GRAVITY EFFECT OF ONE POLYGONAL LAMINA
C NLIST IS THE NUMBER OF COORDINATES
C XLIST ARE THE X COORDS. YLIST ARE THE Y COORDS.
C Z IS THE DEPTH TO THE LAMINA
C TOTEQ IS THE RESULT
      DIMENSION NLIST(40),XLIST(40,40),YLIST(40,40)
      TOTEQ=0.0
      XI=XLIST(NC,1)
      YI=YLIST(NC,1)
      IF(XI.EQ.0..AND.YI.EQ.0.) XI=-DPX/10000.
      RI=SQRT(XI**2+YI**2)
C THIS LOOP TAKES SUCCESSIVE PAIRS OF COORDS AROUND POLYGON  AND
C        MAKES NEW X1,Y1 EQUAL TO OLD X2,Y2 ETC.
      NLISTN=NLIST(NC)
      NLN=NLISTN+1
      DO 4000 IVP=2,NLN
C  IF IVP EXCEEDS NLIST, THEN SET IVP=1
      IF(IVP-NLISTN) 3000,3000,2999
 2999 XIP1=XLIST(NC,1)
      YIP1=YLIST(NC,1)
      IF(XIP1.EQ.0..AND.YIP1.EQ.0.) XIP1=-DPX/10000.
      GO TO 3001
 3000 XIP1=XLIST(NC,IVP)
      YIP1=YLIST(NC,IVP)
      IF(XIP1.EQ.0..AND.YIP1.EQ.0.) XIP1=-DPX/10000.
 3001 RIP1=SQRT(XIP1**2+YIP1**2)
      RIIP1=SQRT((XI-XIP1)**2+(YI-YIP1)**2)
      COMFR1=(XI-XIP1)/RIIP1
      COMFR2=(YI-YIP1)/RIIP1
      PI=COMFR2*XI-COMFR1*YI
      QI=COMFR1*XI/RI+COMFR2*YI/RI
      FI=COMFR1*XIP1/RIP1+COMFR2*YIP1/RIP1
      PPZ=SQRT(PI**2+Z**2)
      IF(PI) 3500,3600,3600
 3500 QI=-QI
      FI=-FI
 3600 EQA=(XI*XIP1+YI*YIP1)/RI/RIP1
      IF(EQA.GT.1.0) EQA=1.0
      EQB=Z*QI/PPZ
      EQC=Z*FI/PPZ
      IF(YI*XIP1-XI*YIP1) 3700,3800,3800
 3700 AREQA=-ACOS(EQA)
      GO TO 3877
 3800 AREQA=ACOS(EQA)
 3877 AREQB=ASIN(EQB)
      AREQC=ASIN(EQC)
      EQN=AREQA-AREQB+AREQC
      TOTEQ=TOTEQ+EQN
C   GET NEW NORMALIZED COORDINATES FROM OLD
      XI=XIP1
      YI=YIP1
      RI=RIP1
 4000 CONTINUE
      RETURN
      END
```

PROGRAM LISTING

PLATE I



PROGRAM FLOW CHART

Record No. 1975/4

G29-218A





Plan view of sphere and contours

• theoretical gravity value
x computed response



0 100 200
Metres

SPHERE MODEL

PLATE 2

Record No.1975/4

G29-219A

Record No 1973/4

DATA SHEET FOR SPHERE EXAMPLE

ITEM	NCONT	DENS	PX	PY	DPX	DPY	NPTS	NLINES	DIST
FORMAT	I 2	F8.4	F10.2	F10.2	F10.2	F10.2	I10	I 10	F10.2
PUNCHED	5	00.5000	-1000.00	0.00	200.00	0.00	11	1	0.00

FORMAT	I 2	I 2	I 2	I 2	I 2	I2 →	I 2	no further values allowed
PUNCHED	1	8	8	8	1	NLIST		

FORMAT	F6.1	F6.1	F6.1	F6.1	F6.1	F 6.1	F6.1	→	F6.1	further values on new card.
PUNCHED	200.0	300.0	500.0	700.0	800.0			← DEPTH		

FORMAT	F6.1	F6.1	F6.1	F6.1	F6.1	F6.1	F6.1	F6.1	→		F6.1	
PUNCHED	-225.0	-160.0	0.0	160.0	225.0	160.0	0.0	-160.0	← XLIST(2)	CONTOUR NO.2.		Further values on new card
..	0.0	160.0	225.0	160.0	0.0	-160.0	-225.0	-160.0	← YLIST(2)			ditto
..	-310.0	-220.0	0.0	220.0	310.0	220.0	0.0	-220.0	← XLIST(3)	CONTOUR NO.3.		ditto
..	0.0	220.0	310.0	220.0	0.0	-220.0	-310.0	-220.0	← YLIST (3)			ditto
..	-225.0	-160.0	0.0	160.0	225.0	160.0	0.0	-160.0	← X LIST (4)	CONTOUR NO.4.		ditto
..	0.0	160.0	225.0	160.0	0.0	-160.0	-2250	-150.0	← Y LIST (4)			ditto

(NO YLIST OR YLIST FOR CONTOURS 1 AND 5 AS THESE ARE SINGLE POINTS)

PLATE 3



DECK STRUCTURE

17